Pages

12 August 2011

Neural Network (Part 4): The Neural Network class

The Neural Network

Finally, we made it to the Neural Network class. This is the top level class that manages the communication between the layers. This class provides the necessary glue to hold all the layers together. It also controls the flow of information forwards and backwards from layer to layer. Lets have a look at a simple 2 layer neural network.

The above neural network starts with 2 inputs in Layer1 and finishes with 2 outputs in Layer 2.

This Neural Network structure is very modular in design. Layers can be added quite easily without affecting the functionality of the neural network code. The only layer that is treated a little bit differently from other layers in the neural network, is the last layer. However, any/all preceding layers are treated exactly the same. And the differences in the last layer, only really come into effect in Neural Network training.

When building my network, I just need to make sure that the number of outputs (or neurons) in the current layer match the number of connections in the next. If I wanted to create a neural network, that accepted 3 input signals from the outside world, to process them with 10 neurons, but only wanted to output one result... I could build the neural network in the following way.

addLayer(3,10) : This first layer would have 3 layerINPUTs and 10 neurons

addLayer(10,1) : This second (and last) layer would have 10 layerINPUTs and 1 neuron

The neural network class would manage these two layers, to ensure that information was passed from one to the other seemlessly. Here is the code for the Neural Network class that would make it possible:

/* -------------------------------------------------------------------------------------
The Neural Network class is a container to hold and manage all the layers
------------------------------------------------------------------------------------- */class NeuralNetwork{
Layer[] layers = {};
float[] arrayOfInputs={};
float[] arrayOfOutputs={};
float learningRate;
float networkError;
float trainingError;
int retrainChances=0;
NeuralNetwork(){
/* the default learning rate of a neural network is set to 0.1, which can changed by the setLearningRate(lR) function. */
learningRate=0.1;
}
/* Function to add a Layer to the Neural Network */void addLayer(int numConnections, int numNeurons){
layers = (Layer[]) append(layers, new Layer(numConnections,numNeurons));
}
/* Function to return the number of layers in the neural network */int getLayerCount(){
return layers.length;
}
/* Function to set the learningRate of the Neural Network */void setLearningRate(float tempLearningRate){
learningRate=tempLearningRate;
}
/* Function to set the inputs of the neural network */void setInputs(float[] tempInputs){
arrayOfInputs=tempInputs;
}
/* Function to set the inputs of a specified layer */void setLayerInputs(float[] tempInputs, int layerIndex){
if(layerIndex>getLayerCount()-1){
println("NN Error: setLayerInputs: layerIndex=" + layerIndex + " exceeded limits= " + (getLayerCount()-1));
} else {
layers[layerIndex].setInputs(tempInputs);
}
}
/* Function to set the outputs of the neural network */void setOutputs(float[] tempOutputs){
arrayOfOutputs=tempOutputs;
}
/* Function to return the outputs of the Neural Network */float[] getOutputs(){
return arrayOfOutputs;
}
/* Function to process the Neural Network's input values and convert them to an output pattern using ALL layers in the network */void processInputsToOutputs(float[] tempInputs){
setInputs(tempInputs);
/* Check to make sure that the number of NeuralNetwork inputs matches the Neuron Connection Count in the first layer. */if(getLayerCount()>0){
if(arrayOfInputs.length!=layers[0].neurons[0].getConnectionCount()){
println("NN Error: processInputsToOutputs: The number of inputs do NOT match the NN");
exit();
} else {
/* The number of inputs are fine : continue */for(int i=0; i<getLayerCount(); i++){
/*Set the INPUTs for each layer: The first layer gets it's input data from the NN whereas the 2nd and subsequent layers get their input data from the previous layer's actual output. */if(i==0){
setLayerInputs(arrayOfInputs,i);
} else {
setLayerInputs(layers[i-1].actualOUTPUTs, i);
}
/* Now that the layer has had it's input values set, it can now process this data, and convert them into an output using the layer's neurons. The outputs will be used as inputs in the next layer (if available). */
layers[i].processInputsToOutputs();
}
/* Once all the data has filtered through to the end of network, we can grab the actualOUTPUTs of the LAST layer
These values become or will be set to the NN output values (arrayOfOutputs), through the setOutputs function call. */
setOutputs(layers[getLayerCount()-1].actualOUTPUTs);
}
}else{
println("Error: There are no layers in this Neural Network");
exit();
}
}
/* Function to train the entire network using an array. */void trainNetwork(float[] inputData, float[] expectedOutputData){
/* Populate the ENTIRE network by processing the inputData. */
processInputsToOutputs(inputData);
/* train each layer - from back to front (back propagation) */for(int i=getLayerCount()-1; i>-1; i--){
if(i==getLayerCount()-1){
layers[i].setDeltaError(expectedOutputData);
layers[i].trainLayer(learningRate);
networkError=layers[i].getLayerError();
} else {
/* Calculate the expected value for each neuron in this layer (eg. HIDDEN LAYER) */for(int j=0; j<layers[i].getNeuronCount(); j++){
/* Reset the delta error of this neuron to zero. */
layers[i].neurons[j].deltaError=0;
/* The delta error of a hidden layer neuron is equal to the SUM of [the PRODUCT of the connection.weight and error of the neurons in the next layer(eg OUTPUT Layer)]. Connection#1 of each neuron in the output layer connect with Neuron#1 in the hidden layer */for(int k=0; k<layers[i+1].getNeuronCount(); k++){
layers[i].neurons[j].deltaError += (layers[i+1].neurons[k].connections[j].weight * layers[i+1].neurons[k].deltaError);
}
/* Now that we have the sum of Errors x weights attached to this neuron.
We must multiply it by the derivative of the activation function. */
layers[i].neurons[j].deltaError *= (layers[i].neurons[j].neuronOutputValue * (1-layers[i].neurons[j].neuronOutputValue));
}
/* Now that you have all the necessary fields populated, you can now
Train this hidden layer and then clear the Expected outputs, ready for the next round */
layers[i].trainLayer(learningRate);
layers[i].clearExpectedOUTPUT();
}
}
}
/* Function to train the entire network, using an array of input and expected data within an ArrayList */void trainingCycle(ArrayList trainingInputData, ArrayList trainingExpectedData, Boolean trainRandomly){
int dataIndex;
/* re-initialise the training Error with every cycle */
trainingError=0;
/* Cycle through the training data either randomly or sequentially */for(int i=0; i<trainingInputData.size(); i++){
if(trainRandomly){
dataIndex=(int) (random(trainingInputData.size()));
} else {
dataIndex=i;
}
trainNetwork((float[]) trainingInputData.get(dataIndex),(float[]) trainingExpectedData.get(dataIndex));
/* Use the networkError variable which is calculated at the end of each individual training session to calculate the entire trainingError. */
trainingError+=abs(networkError);
}
}
/* Function to train the network until the Error is below a specific threshold */void autoTrainNetwork(ArrayList trainingInputData, ArrayList trainingExpectedData, float trainingErrorTarget, int cycleLimit){
trainingError=9999;
int trainingCounter=0;
/* cycle through the training data until the trainingError gets below trainingErrorTarget (eg. 0.0005) or the training cycles have exceeded the cycleLimit variable (eg. 10000). */while(trainingError>trainingErrorTarget && trainingCounter<cycleLimit){
/* re-initialise the training Error with every cycle */
trainingError=0;
/* Cycle through the training data randomly */
trainingCycle(trainingInputData, trainingExpectedData, true);
/* increment the training counter to prevent endless loop */
trainingCounter++;
}
/* Due to the random nature in which this neural network is trained. There may be occasions when the training error may drop below the threshold. To check if this is the case, we will go through one more cycle (but sequentially this time), and check the trainingError for that cycle. If the training error is still below the trainingErrorTarget, then we will end the training session. If the training error is above the trainingErrorTarget, we will continue to train. It will do this check a Maximum of 9 times. */if(trainingCounter<cycleLimit){
trainingCycle(trainingInputData, trainingExpectedData, false);
trainingCounter++;
if(trainingError>trainingErrorTarget){
if (retrainChances<10){
retrainChances++;
autoTrainNetwork(trainingInputData, trainingExpectedData,trainingErrorTarget, cycleLimit);
}
}
} else {
println("CycleLimit has been reached. Has been retrained " + retrainChances + " times. Error is = " + trainingError);
}
}
}

The neural network constructor: NeuralNetwork() - automatically sets the learning rate to 0.1. Other than that, the Neural Network object is an empty shell waiting to be filled. The main setup function of the Neural Network class is the addLayer() function, which adds a layer with a specified number of connections and neurons.

Here are the functions of the Neural Network (NN) class:

addLayer() : adds a layer to the NN (from left to right)

getLayerCount() : returns the number of layers in the NN

setLearningRate(): sets the learning Rate to a specified value.

setInputs(): sets the inputs of the NN, and become the layerINPUTs of the 1st layer

setLayerInputs(): set the inputs of a specified layer

setOutputs(): set the outputs of the NN= same as the actualOUTPUTs of the last layer.

getOutputs(): returns the outputs of the NN

processInputsToOutputs(): Converts the NN inputs into outputs by feeding through layers.

trainNetwork(): uses a training set to train the neural network (using an Array).

trainingCycle(): uses a training set to train the neural network (using an ArrayList).

autoTrainNetwork(): cycles through the training data until a specific condition is met

Ok - so we now have all of the structural components required to build a feed forward neural network. And here is how you would create it. Let us build a neural network that accepts data from 4 sensors and has 3 layers . The layers will have 6, 8 and 2 neurons respectively...

NeuralNetwork NN = new NeuralNetwork();

NN.addLayer(4,6);

NN.addLayer(6,8);

NN.addLayer(8,2);

Perfect, we have just created an entire neural network, with randomised weights and biases etc etc.

Unfortunately, the neural network is not that useful at the moment. Before we can even start to use it, we need to put it through school. We need to teach it.

Try to imagine a colour that you have never seen before. Hard isn't it ? That is how the Neural Network feels. So before you can get it to make any sort of classifications, you have to show it some examples of what you are looking for. Once the neural network can make sense of your examples, you use a "validation set" to put it through it's paces. If it performs well, then you are good to go, otherwise it is back to school until it can pass the test. My neural network doesn't have a validation set at the moment, but it seems to work quite well without it. This statement is not entirely true. I tend to validate it using the training data, but maybe in future I will fix it up to use a proper validation set.

Hi Andrew,The WHILE loop (before the line in question) will cycle through randomly until the cycle count exceeds 10000 cycles or until the training error drops below 0.005 (which ever comes first). The IF statement after the WHILE LOOP checks to see the cycle count is responsible for breaking the WHILE LOOP conditions. If the cycle count has exceeded the maximum cycle count (or cycleLimit), then it will display a message to that effect. On the other hand, if there are still some cycles remaining (trainingCounter<cycleLimit), it means that the reason the WHILE loop has exited, was due to a trainingError that has dropped below 0.005. Therefore, it will cycle through one more time, however on this occasion, will cycle through sequentially (rather than randomly). At the end of this cycle, a new trainingError will be set, and will be tested against the trainingTarget of 0.005. If on this occasion it is now GREATER than 0.005, then it will need to go back and try again (by training randomly once again). Hence the call to autoTrainNetwork. It will only have 9 chances to get it right, before it gives up.

I hope this explains it a bit better. The main aim of this part of code was to randomise the training. There are probably better ways of doing this, but the way I did it made sense to me at the time.

In my system there are two outputs but at the time of training sometimes one of the outputs will not be able to provided. However i want the NN produce output for that output too. Is this possible? In this case how can we apply the single data?